ABSTRACT
In the compute block of the PROC REPORT step variables can be both created and used. They may be on the incoming data table, they may be computed in the REPORT step, they can be statistical summaries, and they may even be artificially created by the REPORT process. In the DATA step each variable has a name and that variable’s name is used whenever we want to address a particular value on the Program Data Vector. In the REPORT step things are not so simple. The origins and use of a variable will determine how it is to be named in a compute block.

If you do not know how to properly address a variable or report item in the compute block, the compute block will often fail. This paper will discuss the various variable types, how they are used, and how they are addressed in the compute block. Once introduced the rules are fairly easy, and a full understanding of the naming conventions and how they are applied - “use what name when”, will go a long way to allowing you to create more complex and successful compute blocks in PROC REPORT.

KEYWORDS
PROC REPORT, Compute Block, Computed variables, Composite names, Report items

INTRODUCTION
The compute block is an extremely valuable tool within PROC REPORT, and since it supports the use of many of the programming statements and functions of the DATA step, it can be used to create columns and manipulate data. To fully use this power we need to be able to access the incoming variables, as well as, columns and statistics created in the step itself. These columns and variables have names and as the naming conventions are not as straight forward as in the DATA step, a more complete understanding of how to access them is required.

The following simple compute block writes the value of the variable REGION using the format $REGNAME. This compute block is referencing the report item, REGION, which is a variable on the incoming data table.

``` Sas
compute before region;
   line @3 region $regname8.;
endcomp;
```

There are four ways to reference a variable in a compute block, and you can reference any report item that forms a column, even columns that are not printed. In the LINE statement of the above compute block, the variable REGION, which is also a report item and a GROUP variable, is referenced directly (explicitly) by name.

You can reference report items:

- **explicitly by name**
The variable name can be used directly, as in the above example, when the variable has a define type of GROUP, ORDER, COMPUTED, or DISPLAY. Temporary variables, variables that are created and only used in a compute block, are also always addressed directly by variable name.

- **using a compound name**
Compound variable names are needed, when a variable with an ANALYSIS define type has been used to calculate a statistic. The compound name is a combination of the variable name and the statistic that it
has been used to calculate. The general form is `variablename.statistic`, and an example of a compound name might be:

```
wt.mean
```

- **by specifying an alias**
  An alias can be specified in the COLUMN statement when you want to use a single analysis variable in more than one way - generally to calculate more than one statistic. Aliases are named directly in a compute block. The following COLUMN statement generates a series of aliases for the HT analysis variable.

```plaintext
columns region ht
   ht=htmin ht=htmax
   ht=htmean ht=htmedian;
```

In the compute block the alias is addressed directly as in:

```plaintext
compute after;
   line @3 'Minimum height is ' htmin 6.1;
endcomp;
```

- **indirectly using the absolute column number**
  Sometimes as the report is constructed a given column may not have a specific name. This is especially the case when a variable, with the define type of ACROSS, creates a series of columns. These, and indeed any column in the report, can be referenced by using the column number as an indirect column reference. This pseudo variable name is always of the form `_Cxx_`

where the `xx` is the column number as read from left to right on the report. The column count includes any columns that will ultimately not print e.g. those columns defined with NOPRINT or NOZERO.

The REPORT step also creates an automatic temporary variable named `_BREAK_`. This variable is addressed directly and is available during the execution of the compute block.

**DIRECT VARIABLE NAME REFERENCE**
When a variable has the define type of DISPLAY, that variable’s name is used explicitly in compute blocks. This is shown in the following example which calculates a computed column.

The Body Mass Index, BMI, is a rough measure of health, and for most adults a BMI between 18.5 and 24.9 is generally considered to be in the normal range. The following example calculates the BMI for the students in the SASHELP.CLASS data set.

```plaintext
title1 'Extending Compute Blocks';
title2 'Using Variable Names';
proc report data=sashelp.class nowd;
column name weight height BMI;
define name / display;
define weight / display;
define height / display;
define bmi / computed format=4.1 'BMI';
compute bmi;
   bmi = weight / (height*height) * 703;
endcomp;
run;
```
Extending Compute Blocks
Using Variable Names

<table>
<thead>
<tr>
<th>Name</th>
<th>Weight</th>
<th>Height</th>
<th>BMI</th>
</tr>
</thead>
<tbody>
<tr>
<td>Alfred</td>
<td>112.5</td>
<td>69</td>
<td>16.6</td>
</tr>
<tr>
<td>Alice</td>
<td>84</td>
<td>56.5</td>
<td>18.5</td>
</tr>
<tr>
<td>Barbara</td>
<td>98</td>
<td>65.3</td>
<td>16.2</td>
</tr>
<tr>
<td>Carol</td>
<td>102.5</td>
<td>62.8</td>
<td>18.3</td>
</tr>
<tr>
<td>Henry</td>
<td>102.5</td>
<td>63.5</td>
<td>17.9</td>
</tr>
<tr>
<td>James</td>
<td>83</td>
<td>57.3</td>
<td>17.8</td>
</tr>
<tr>
<td>Jane</td>
<td>84.5</td>
<td>59.8</td>
<td>16.6</td>
</tr>
<tr>
<td>Janet</td>
<td>112.5</td>
<td>62.5</td>
<td>20.2</td>
</tr>
<tr>
<td>Jeffrey</td>
<td>84</td>
<td>62.5</td>
<td>15.1</td>
</tr>
<tr>
<td>John</td>
<td>99.5</td>
<td>59</td>
<td>20.1</td>
</tr>
<tr>
<td>Joyce</td>
<td>50.5</td>
<td>51.3</td>
<td>13.5</td>
</tr>
<tr>
<td>Judy</td>
<td>90</td>
<td>64.3</td>
<td>15.3</td>
</tr>
<tr>
<td>Louise</td>
<td>77</td>
<td>56.3</td>
<td>17.1</td>
</tr>
<tr>
<td>Mary</td>
<td>112</td>
<td>66.5</td>
<td>17.8</td>
</tr>
<tr>
<td>Philip</td>
<td>150</td>
<td>72</td>
<td>20.3</td>
</tr>
<tr>
<td>Robert</td>
<td>128</td>
<td>64.8</td>
<td>21.4</td>
</tr>
<tr>
<td>Ronald</td>
<td>133</td>
<td>67</td>
<td>20.8</td>
</tr>
<tr>
<td>Thomas</td>
<td>85</td>
<td>57.5</td>
<td>18.1</td>
</tr>
<tr>
<td>William</td>
<td>112</td>
<td>66.5</td>
<td>17.8</td>
</tr>
</tbody>
</table>

In this example the two variables \texttt{HEIGHT} and \texttt{WEIGHT} have a define type of \texttt{DISPLAY}. When \texttt{DISPLAY} variables are used in a compute block they are referenced directly.

If these same variables had been designated as \texttt{ANALYSIS} (which is typical, and the default for numeric variables without \texttt{DEFINE} statements), the previous compute block would not have worked (see the same example in the next section using these variables with a define type of \texttt{ANALYSIS}). The LOG would contain the following message, which seems to be wrong and is therefore confusing.

\begin{verbatim}
NOTE: Variable height is uninitialized.
NOTE: Variable weight is uninitialized.
NOTE: Division by zero detected at line 1 column 15.
\end{verbatim}

Of course the variables \texttt{HEIGHT} and \texttt{WEIGHT} do exist and are initialized, but they have different names in the compute block when they have a define type of \texttt{ANALYSIS}.

The following example converts the weight of each class member from pounds to kilograms. Notice that the compute block is associated with a variable with a type of \texttt{DISPLAY}.

```
title1 'Extending Compute Blocks';
title2 'Using Variable Names';
title3 'Converting Pounds to Kilograms';

proc report data=sashelp.class nowd;
column name sex ('Weight' weight);
define name / order 'Name';
define sex / display 'Sex';
define weight / display format=6. 'Kg';
compute weight;
   weight = weight / 2.2;
endcomp;
run;
```

The compute block will execute for each row of the report, and since \texttt{WEIGHT} has a define type of \texttt{DISPLAY} a direct variable reference is used in the compute block to name the variable to be converted.
User defined temporary variables are created in compute blocks and are always addressed explicitly using the variable name. This is demonstrated in the following example that counts observations using the temporary variable CNT.

One of the default features of PROC PRINT is the OBS column. The following example adds this observation counter for PROC REPORT output. In the compute block where the OBS column is computed two variables are used. The variable CNT does not appear on the COLUMN statement and is therefore a temporary variable. Its value will be retained from row to row, consequently a SUM statement is especially useful. The value of CNT, which is the current row number, is assigned to the computed variable OBS.

```sql
title1 'Extending Compute Blocks';
title2 'Using Variable Names';
title3 'Creating an OBS Column';
proc report data=sashelp.class nowd;
column obs name sex weight;
define obs / computed 'Obs' format=3.;
define name / order 'Name';
define sex / display 'Sex';
define weight / display format=6.2 'Weight';
compute obs;
   cnt + 1;
   obs = cnt;
endcomp;
run;
```

The resulting table becomes:

```
Extending Compute Blocks
Using Variable Names
Creating an OBS Column

Obs Name  x  Weight
1  Alfred M  112.50
2  Alice  F  84.00
3  Barbara F  98.00
4  Carol  F  102.50
... Portions of the report are not shown ...
```

Notice that the computed variable OBS and the temporary variable CNT are both addressed specifically.
FYI - The problem of the generation of the OBS column is specifically addressed by Pass (2000).

**COMPOUND VARIABLE NAMES**

In the first example of the previous section, the BMI is calculated for the students in SASHELP.CLASS. In that example the variables WEIGHT and HEIGHT received a define type of DISPLAY and the variables were address explicitly using their names. When the define type of ANALYSIS is used, the variables cannot be address explicitly, instead the use of compound names is required.

```sas
title1 'Extending Compute Blocks';
title2 'Using Compound Variable Names';
proc report data=sashelp.class nowd;
column name weight height BMI;
define name   / display;
define weight / analysis;
define height / analysis;
define bmi    / computed format=4.1 'BMI';
compute bmi;
   bmi = weight.sum / (height.sum*height.sum) * 703;
endcomp;
run;
```

A compound name is always going to be a combination of the variable name and a statistic. When a statistic has not been specified, the statistic will be the SUM.

It is because HEIGHT and WEIGHT are analysis variables, that compound names are required in the compute block. Whereas the following assignment statement worked in the first example of the previous section, it would not have worked here, because compound names are not being used.

```sas
compute bmi;
   bmi = weight / (height*height) * 703;
endcomp;
```

When statistics are specified for analysis variables they must also appear as part of the compound name if that variable is used in a compute block. The following example converts the mean weight of class members from pounds to kilograms. Notice that the compute block is associated with a variable with a type of ANALYSIS, an this means that compound variable names will be used in the compute block.

```sas
title1 'Extending Compute Blocks';
title2 'Using Variable Names';
title3 'Converting Pounds to Kilograms';
proc report data=sashelp.class nowd;
column sex ('Weight' weight);
define sex    / group 'Sex' format=$3.;
define weight / analysis mean format=6.1 'Kg';
compute weight;
```
weight.mean = weight.mean / 2.2;
endcomp;
run;

Notice that the name of the compute block itself is NOT a compound name. The issues with the naming of variables within a compute block refer only to the naming of the variables WITHIN a compute block.

**USING AN ALIAS AS A COLUMN REFERENCE**

A variable alias is created on the COLUMN statement, and once created, each alias can be associated with a DEFINE statement. When aliases are used in a compute block they are addressed using the alias name and not the original variable’s name.

In the following example the alias WTMAX, which is created with the specification \( WT = WTMAX \) in the COLUMN statement, allows the you to use the report item WT in two different DEFINE statements. This alias can then be used directly in the compute block.

```plaintext
title1 'Using an Alias';
proc report data=sashelp.class nowd;
  column sex ('Weight in Pounds' weight weight=wtmax wt_range);
  define sex      / group format=$3.;
  define weight   / analysis min 'Min' format=5.1;
  define wtmax    / analysis max 'Max' format=5.1;
  define wt_range / computed 'Range' format=5.1;
  compute wt_range;
    wt_range = wtmax - weight.min;
  endcomp;
run;
```

Notice the use of the compound name `weight.min` in the compute block assignment statement 1. Because WTMAX is an alias, the following assignment statement would not work:

```plaintext
wt_range = wtmax.max - wt.min;
```

Nor could we ignore the compound name for the analysis variable WT. The following assignment statement would also fail.

```plaintext
wt_range = wtmax - wt;
```

**USING ABSOLUTE COLUMN NUMBER REFERENCES**

Derived columns, such as those created using the ACROSS define type, will not have a column name on the COLUMN statement. When these columns are used in a compute block, an indirect column reference is required. Since this form of column referencing uses the actual number of the column it is also known as absolute column number referencing.

In the following example the ratio of the mean weight of the two genders is calculated. Since we want the weight of the females to be the numerator we use `_c3_ / `_c2_` as the ratio.

```plaintext
title1 'Using Indirect Column References';
```
proc format;
  value $gender
    'M' = 'Male'
    'F' = 'Female';
run;

proc report data=sashelp.class nowd split='*';
column age ('Mean Weight*in Pounds' sex,weight ratio);
define age    / group width=10 'Age';
define sex    / across 'Gender'
  format=$gender.
  order=data;
define weight / analysis mean format=6. ' ';
define ratio  / computed format=6.3 'Ratio*F/M '
  rbreak after / skip summarize;
compute ratio;
  ratio = _c3_ / _c2_;
endcomp;
run;

We verify that the weight for males is in the second column (_C2_) and that the weight of the females is in the third column (_C3_).

Since the column numbers of the table are used, the programmer must have a certain knowledge of the final table before selecting the numbers used in the indirect reference. In the previous example we know that in the incoming data table the first observation is for a male patient and that the ORDER= option has been set to DATA. Changing the DEFINE statement for SEX to use ORDER=FORMATTED

```sas
define sex / across 'Gender'
  format=$gender. order=formatted;
```

would place FEMALES in column 2 and the assignment statement for RATIO would be:

```sas
compute ratio;
  ratio = _c2_ / _c3_;
endcomp;
```
USING THE AUTOMATIC TEMPORARY VARIABLE _BREAK_
As PROC REPORT executes, the data is summarized and held in memory before being transferred to the report itself. One of the columns in this intermediate summary is the automatic variable _BREAK_. This column is used to identify lines were generated as a result of grouping variables, BREAK, or RBREAK statements.

Adding an OUT= option to the PROC statement in the previous example allows us to get a glimpse of this temporary table.

```latex
proc report data=sashelp.class nowd split='*' out=temptbl;
   . . . . statements not shown . . . .
run;

proc print data=temptbl;
run;
```

The resulting data table is shown below. _BREAK_ is a character variable and can be addressed directly in the compute block. It has missing values for the detail rows of the report, however for the line that is generated for the report break, the variable _BREAK_ is equal to '_RBREAK_'. Notice that AGE is missing for the summary line.

<table>
<thead>
<tr>
<th>Obs</th>
<th>Age</th>
<th><em>C2</em></th>
<th><em>C3</em></th>
<th>ratio</th>
<th><em>BREAK</em></th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>11</td>
<td>85.00</td>
<td>50.500</td>
<td>0.59412</td>
<td></td>
</tr>
<tr>
<td>2</td>
<td>12</td>
<td>103.50</td>
<td>80.750</td>
<td>0.78019</td>
<td></td>
</tr>
<tr>
<td>3</td>
<td>13</td>
<td>84.00</td>
<td>91.000</td>
<td>1.08333</td>
<td></td>
</tr>
<tr>
<td>4</td>
<td>14</td>
<td>107.50</td>
<td>96.250</td>
<td>0.89535</td>
<td></td>
</tr>
<tr>
<td>5</td>
<td>15</td>
<td>122.50</td>
<td>112.250</td>
<td>0.91633</td>
<td></td>
</tr>
<tr>
<td>6</td>
<td>16</td>
<td>150.00</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>7</td>
<td>.</td>
<td>108.95</td>
<td>90.111</td>
<td>0.82709</td>
<td><em>RBREAK</em></td>
</tr>
</tbody>
</table>

The _BREAK_ variable can be very valuable when using logic statements within compute blocks that need to detect group boundaries.

SUMMARY
Depending on the type of variable and how it is used there are different ways to name variables within the PROC REPORT’s compute block:

- Direct reference for DISPLAY variables, computed variables, variable aliases, and temporary variables
- Compound names for ANALYSIS variables and their associated statistics
- Indirect naming of columns by report column number

There is also the special temporary variable, _BREAK_, which is available for use in programming statements within the compute block. This variable is also addressed directly by name.

ABOUT THE AUTHOR
Art Carpenter’s publications list includes four books, and numerous papers and posters presented at SUGI and other user group conferences. Art has been using SAS® since 1976 and has served in various leadership positions in local, regional, national, and international user groups. He is a SAS Certified Professional™ and through California Occidental Consultants he teaches SAS courses and provides contract SAS programming support nationwide.
AUTHOR CONTACT
Arthur L. Carpenter
California Occidental Consultants
P.O. Box 430
Vista, CA 92085-0430
(760) 945-0613
art@caloxy.com
www.caloxy.com

REFERENCES

TRADEMARK INFORMATION
SAS, SAS Certified Professional, and all other SAS Institute Inc. product or service names are registered trademarks of SAS Institute, Inc. in the USA and other countries. © indicates USA registration.